
6

變數與陣列
6-1變數與資料型態
　　設計程式的目的，無非是要用程式來幫我們完成一些資料的運算與處理，不過，在運算的過程中，不論待處理或已處理的資料都必須想辦法將其暫存起來。最簡單的方式就是使用「變數」（Variable）來完成資料暫存的功能。

所以，變數其實就是程式中用來儲存資料的預留記憶體。

6-1-1變數宣告
使用變數前最好先「宣告」（Declaration），宣告變數的目的就是要配置一塊記憶體供該變數使用，ASP程式中用來宣告變數的VBScript語法如下。

· 語法：


Dim 變數名稱
· 說明：

1. VBScript是屬於鬆散式的描述語言，只提供一種預設的資料型態（Data type），稱為變異型態（Variant）。

2. Variant可以用來處理任何型態的資料，包括整數、字串、浮點數 ... 等，因此VBScript在宣告時並不須特別指定資料型態。

3. 如果在程式中需要以特定的資料型態來運作，請參照下一小節的「資料型態轉換」。
4. VBScript在使用變數前其實並不一定要事先宣告，但是若能養成每個變數在使用前都宣告的好習慣，對於程式的除錯將有莫大的幫助。

· 變數命名規則
　　為變數取名稱時，必須遵循如下的命名規則：

1. 變數名稱的啟始字元（第一個字元）必須為英文字母（a、b … z）。

2. 變數名稱中除了英文字母（a、b … z），數字（0, 1, … 9）和底線符號（_）外，不得含有標點符號、空白等特殊字元。

3. 變數名稱不可與保留字相同（例如If、For、While ... 等）。

4. 以下列舉數個合法與不合法的變數名稱供讀者參考：

· 合法的變數名稱如：

a

a2

student

student_number

school_name
· 不合法的變數名稱如：

2


（不得使用數字為起始字元）

2a


（不得使用數字為起始字元）

%abc


（不得使用特殊符號 % 為起始字元）

a*b


（不得含有運算符號 *）
a<


（不得含有比較運算子 <）

student-number
（不得含有運算符號 -）

school name
（不得含有空白）

For


（不得使用保留字）
6-1-2常數宣告

變數在程式執行過程中，可以隨時加已改變，而常數（Constant）就是恆久不變的數，像圓週率為3.14159即是一例。ASP程式中用來宣告常數的VBScript語法如下。

· 語法


Const 常數名稱 = 常數值
6-1-3資料型態

由於VBScript只提供一種預設的資料型態Variant，因此在ASP程式中便一律使用Variant來處理任何型態的資料，包括整數（Integer）、長整數（Long）、單精度浮點數（Single）、倍精度浮點數（Double）、字串（String）、布林（Boolean）、貨幣（Currency） ... 等。實際上，當資料被存入變數時，Variant型態會自動轉換成符合該資料的真正型態，我們可以用VarType(變數名稱)這個函數來判斷真正的資料型態為何，VarType函數的回傳值所代表的意義如下表所示。
	傳回值
	代號
	說明

	0
	VbEmpty
	未初始化的變數（Empty）

	1
	VbNull
	Null（無有效資料）

	2
	VbInteger
	整數

	3
	VbLong
	長整數

	4
	VbSingle
	單倍精確浮點數

	5
	VbDouble
	雙倍精確浮點數

	6
	VbCurrency
	貨幣

	7
	VbDate
	日期

	8
	VbString
	字串

	9
	VbObject
	物件

	10
	VbError
	錯誤

	11
	VbBoolean
	布林值

	12
	VbVariant
	Variant（只適用於variants中的陣列）

	13
	VbDataObject
	資料存取物件

	14
	VbDecimal
	十進位值

	36
	VbUserDefinedType
	包含使用者自定型態的變數

	17
	VbByte
	位元值

	8192
	VbArray
	陣列


· 範例 6-1（ch06-01.asp）
· 程式功能

本範例說明變數宣告與資料型態轉換。

· 程式碼
<html>

<head>

<title>範例 6-1</title>

</head>
<body>

<h3>本範例說明變數與資料型態</h3>

<hr>

<%

  Dim varX, nulX, intX, strX, bolX

  nulX = Null

  intX = 3

  strX = "林信成"

  bolX = True

  Response.Write "varX 真正的型態為: " & VarType(varX) & "-Empty<br>"

  Response.Write "nulX 真正的型態為: " & VarType(nulX) & "-Null<br>"

  Response.Write "intX 真正的型態為: " & VarType(intX) & "-整數<br>"

  Response.Write "strX 真正的型態為: " & VarType(strX) & "-字串<br>"

  Response.Write "bolX 真正的型態為: " & VarType(bolX) & "-布林<br>"
%>
</body>

</html>
· 執行結果

6-2型態轉換函數
有時候我們會希望能以特定的型態來處理資料，此時可以使用資料型態轉換函數，將資料由預設的資料型態轉換成特定的資料型態。下表簡述可在ASP程式中使用的VBScript型態轉換函數：
	函數名稱
	語法
	說明

	Cint 
	CInt(expression)
	此函數傳回一個轉換為Integer資料型態的運算式。引數expression可以是任何有效的運算式。

	CLng 
	CInt(expression)
	此函數傳回一個轉換為Long資料型態的運算式。引數expression可以是任何有效的運算式。

	CSng
	CSng(expression)
	此函數傳回一個轉換為Single資料型態的運算式。引數expression可以是任何有效的運算式。

	CDb
	CDbl(expression)
	此函數傳回一個轉換為Double資料型態的運算式。引數expression可以是任何有效的運算式。

	CStr
	CStr(expression)
	此函數傳回一個轉換為String資料型態的運算式。引數expression可以是任何有效的運算式。

	CBool
	CBool(expression)
	此函數傳回一個轉換為Boolean資料型態的運算式。引數expression可以是任何有效的運算式。如果 expression 的結果為 0 ，則傳回 False；否則將傳回 True。如果 expression 無法被計算成數值，將會產生執行階段錯誤。

	CByte
	CByte(expression)
	此函數傳回一個轉換為Byte資料型態的運算式。引數expression可以是任何有效的運算式。

	CCur
	CCur(expression)
	此函數傳回一個轉換為Currency資料型態的運算式。引數expression可以是任何有效的運算式。

	CDate
	CDate(date)
	此函數傳回一個轉換為Date資料型態的運算式。引數date可以是任何有效的日期運算式。您可以使用 IsDate 函數來檢查date是否可以被轉換成日期或時間。而CDate接受短式日期和時間文字以及日期/時間有效範圍內的適當數值。當轉換一個數字成為日期時，是將整數部分轉換成日期；任何數字中的小數部分，將轉換為從午夜起算的時間。


6-3陣列


本節說明陣列之觀念，有些地方必須用到迴圈（Loop），若對迴圈不熟的讀者可先參閱本書第十章，再回頭來閱讀本節。
6-3-1何謂陣列
　　陣列（Array）簡單的說就是：一群資料型態的集合，是相當常見也廣為使用的一種資料結構（Data structure）。為什麼需要陣列呢？我們都知道，電腦最大的用途就是資料處理（Data processing）。當資料輸入電腦時，會在記憶體中配置一個變數來儲存，也就是說若要處理10筆資料，需要10個變數來儲存。現在以一個假想的例子來說明：假設有一函數InputBox可以用來輸入資料，若某班級有十個同學，那麼要輸入十位同學的學號，可以使用10個變數id1, id2, ..., id10來儲存：

id1 = InputBox ("請輸入ID1")

id2 = InputBox("請輸入ID2")

                ...

                ...

                ...

id10 = InputBox ("請輸入ID10")
　　當資料量不大時，這種為每個變數取一個名稱，處理時也是分別處理的方法問題不大。但是，試想當資料量增加為100筆、1000筆...時，這種方式不但非常沒有效率，而且程式也會龐大到無法維護。

　　使用陣列便可以解決這類的問題。

　　例如全校有1000位同學，只要在程式中宣告一個陣列，並配合迴圈就可以完成1000個學號的輸入工作：

Dim id(1000)

For i = 1 To 1000

    id(i) = InputBox ("請輸入Please key in ID" & i)

Next I

　　陣列的觀念類似數學中的矩陣（Matrix）。陣列中每個成員（Member）都是一個變數，稱為陣列元素（Element）或註標變數（Subscript variable）。陣列元素共用一個陣列名稱，而以註標（Subscript）來識別。例如：


[image: image2.wmf] 

 

 

id(0)

 

2312

 

id(1)

 

5234

 

…

 

…

 

id(1000)

 

3715

 

 

陣列元素

 

註標

 

陣列名稱

 

資料

 


　　陣列依其大小是否可改變，可區分為：

· 固定陣列：大小不可任意改變的陣列

· 動態陣列：大小可隨意改變的陣列

6-3-2固定陣列
· 意義

　　所謂固定陣列是指一經宣告之後，大小即無法加以改變的陣列。

· 宣告語法


Dim 陣列名稱(註標)

· 說明

1. 陣列名稱的命名規則與一般變數相同。

2. 註標決定陣列的維數及上下限（大小），預設的下限值為0。
3. 多為陣列，註標請以逗點隔開。
4. 例如：

	宣告
	陣列元素

	Dim A(10)
	A(0) ~ A(10)

	Dim C(2, 3)
	C(0, 0) ~ C(2, 3)


· 範例 6-2（ch06-02.asp）
· 程式功能：

隨機產生10個介於0~100的整數，並將其輸出。
· 程式碼
<html>
<head>

<title>範例 6-2</title>

</head>

<body>

<h3>本範例說明固定陣列的使用</h3>

<hr>
<%

  Dim A(10)

  for i = 1 to 10

    A(i) = CInt(100 * rnd)

    Response.Write "陣列中第 " & i & " 個數為：" & A(i) & "<br>"

  next

%>
</body>
</html>
· 執行結果

· 範例 6-3（ch06-03.asp）
· 程式功能：

　　試設計一程式，隨機產生10個介於0~100的整數，然後將其分別「由大至小」及「由小至大」排序輸出。

· 程式碼：

<html>

<head>

<title>範例 6-3</title>

</head>

<body>

<h3>範例：陣列的排序</h3>

<hr>

<body>

<%

  Dim i, j, tmp, a(10)

  Response.Write "<p>原始資料：</p>"

  Randomize

  For i = 1 To 10

    a(i) = Int(Rnd * 100)

    Response.Write a(i) & ", "

  Next

  Response.Write "<p>由大至小排序結果：</p>"

  For i = 1 To 9

    For j = i + 1 To 10

        If a(i) < a(j) Then

            tmp = a(i)

            a(i) = a(j)

            a(j) = tmp

        End If

    Next

  Next

  For i = 1 To 10

    Response.Write a(i) & ", "

  Next

  Response.Write "<p>由小至大排序結果：</p>"

  For i = 1 To 9

    For j = i + 1 To 10

        If a(i) > a(j) Then

            tmp = a(i)

            a(i) = a(j)

            a(j) = tmp

        End If

    Next

  Next

  For i = 1 To 10

    Response.Write a(i) & ", "

  Next

%>

</body>

</html>

· 執行結果

· 說明

本範例使用泡沫排序法（Bubble sorting）。說明如下：

1. 若是由大至小排序，先將a(1) 與 a(2) 比較，若a(1) < a(2)，則將兩者交換，否則不換。再以同樣的方式，將a(1) 分別與 a(3), a(4) ... a(10) 等比較，則比完一輪後，a(1) 必為10個數中之最大數。

2. 第二輪則以a(2) 為主，分別與a(3), a(4) ... a(10) 比較，則比完第二輪後，a(2) 必為10個數中僅次於a(1)的次大數。

3. 第三輪則以a(3) 為主，分別與a(4), a(5) ... a(10) 比較，則比完第三輪後，a(3) 必為10個數中第3大數。

4. 依以上步驟反覆進行，比完第九輪時，a(9) 必為第9大數，剩下的a(10) 則為最小數。

5. 由小至大排序雷同，請讀者自行揣摩。

6. 排序法中需用到兩數互換的技巧，例如a(i) 與a(j) 要交換，可以使用下列敘述：

tmp = a(i)
a(i) = a(j)

a(j) = tmp
6-3-3動態陣列
· 意義

　　所謂動態陣列是指宣告之後，大小仍可隨時加以改變的陣列。

· 宣告語法

　　動態陣列在宣告時可以先不必指定註標值，而以空刮號 () 表示其大小未定，如下所示：


Dim 陣列名稱()
　　當需要使用時，再以如下敘述指定註標，以確定陣列的大小：


ReDim 陣列名稱(註標)
· 說明

1. 陣列名稱的命名規則與一般變數相同。

2. 註標決定陣列的維數及上下限（大小），預設的下限值為0。
3. 多為陣列，註標請以逗點隔開。

4. 調整陣列維度或大小的ReDim敘述，可置於程序中任何位置，並可反覆出現。
5. 一旦下達ReDim，則該陣列被重置，之前的內容一概消失。
· 範例 6-4（ch06-04.asp）
· 程式功能
改寫範例6-3，以一HTML表單輸入一整數N，而ASP程式則隨機產生N個介於0~100的整數，然後將其「由大至小」及「由小至大」排序輸出。
· HTML表單原始碼

<html>

<head>

<title>範例 6-4</title>

</head>

<body>

<h3>範例：動態陣列的排序</h3>

<hr>

<body>

<form method="post" action="ch06-04.asp">

請輸入 N : <input type="text" name="txtN">

<p>本程式將隨機產生 N 個整數，然後分別<br>

「由大至小」及「由小至大」排序輸出。

</p>

<p>

<input type="submit" value="確定">

<input type="reset" value="清除">

</p>

</form>
</body>

</html>
· ASP程式
<html>

<head>

<title>範例 6-4</title>

</head>

<body>

<h3>範例：動態陣列的排序</h3>

<hr>

<body>

<%

  Dim n, i, j, tmp, a()

  n = CInt(Request.Form("txtN"))

  ReDim a(n)

  Response.Write "<p>原始資料：</p>"

  Randomize

  For i = 1 To n

    a(i) = Int(Rnd * 100)

    Response.Write a(i) & ", "

  Next

  Response.Write "<p>由大至小排序結果：</p>"

  For i = 1 To n-1

    For j = i + 1 To n

        If a(i) < a(j) Then

            tmp = a(i)

            a(i) = a(j)

            a(j) = tmp

        End If

    Next

  Next

  For i = 1 To n

    Response.Write a(i) & ", "

  Next

  Response.Write "<p>由小至大排序結果：</p>"

  For i = 1 To n-1

    For j = i + 1 To n

        If a(i) > a(j) Then

            tmp = a(i)

            a(i) = a(j)

            a(j) = tmp

        End If

    Next

  Next

  For i = 1 To n

    Response.Write a(i) & ", "

  Next

%>

</body>

</html>
· 執行結果
· HTML表單執行結果

[image: image5.png]HB - BRRESTRPERF

E oS HC

FAEF AT N EE) > SRR
TEAZEDN B TEVNER PR

e





· ASP程式執行結果

6-4 Array函數
· 用途

產生一個陣列。

· 語法


Array(arglist)
· 說明

arglist引數是必要的，而且是一串以逗號隔開的值，這些值是陣列的元素。如果沒有指定引數，則會建立長度為零的陣列。

· 範例 6-5（ch06-05.asp）
· 程式功能

試寫一個抽獎程式，以亂數隨機產生點數0~6，分別對應如下獎項：

0 - 槓龜獎：銘謝惠顧


1 - 頭獎：機車一部



2 - 貳獎：電視一台


3 - 參獎：數位相機一部



4 - 肆獎：遙控車一台


5 - 伍獎：背包一個


6 - 陸獎：咖啡禮盒一盒
· 程式碼
<html>

<head>

<title>範例 6-5</title>

</head>

<body>

<h2>範例：抽獎程式</h2>

<hr>

<%

  Dim a, i

  a = Array("槓龜獎（銘謝惠顧）", _



"頭獎（機車一部）", _



"貳獎（電視一台）", _



"參獎（數位相機一部）", _



"肆獎（遙控車一台）", _



"伍獎（背包一個）", _



"陸獎（咖啡禮盒一盒）")

  Randomize

  i = CInt(6 * rnd)

  Response.Write "<p>您抽中點數："  & i & "</p>"

  Response.Write "恭禧您獲得："  & a(i)

%>

</body>

</html>
· 執行結果

習題
1. 試設計一程式，計算下列2階行列式的值：

[image: image8.bmp]
2. 試設計一程式，求解下列二元一次方程式：

[image: image9.bmp]
3. 試設計一程式，計算下列矩陣的乘積：

4. 假設某班級有10學生，每人有國文、英文、數學三科成績，試設計一程式，計算下列各值：

(a)
每一位同學的總成績及平均成績

(b)
國文科全班平均成績

(c)
英文科全班平均成績

(d)
數學科全班平均成績

(e)
全班總平均成績

