
15

ASP入門

15-1認識ASP

ASP（Active Server Page）是由微軟所發展的一套建構雙向互動式網頁的技術。由於HTML是一種「標注語言」（Markup Language），擅長版面編排及展示，不像一般「程式語言」（Programming Language）具備流程控制、條件分支、迴圈結構、輸出入互動 ... 等能力。於是為了使WWW網頁具備與使用者互動的功能，乃有各種內嵌於HTML中的「描述式語言」（Scripting Language）被發展出來，如前述的JavaScript和VBScript。但是這些描述式語言的用途僅止於應用於前端與使用者互動的程式設計上，至於要完成與後端伺服器的互動，便力有未逮。

建立雙向互動式網頁最早的技術當推CGI，目前都還非常普遍的被使用著。CGI 是Common Gateway Interface的縮寫，中文翻譯為「共同閘道介面」，1993年以來廣為WWW所採用。CGI並不是一種程式語言或資料格式，而是一套規範 WWW伺服器和「描述程式」（Script）之間溝通的介面標準。一般都把「遵循CGI介面標準的描述程式」簡稱為「CGI程式」，其主要用途是接收、處理資料。其運作方式為：用戶端將「要求」（Request）傳送到伺服器，伺服器將其轉換成各種「環境變數」（Environment variables）傳送給對應的CGI程式處理，CGI程式經由讀取環境變數而得知客戶端的要求，隨後將處理結果透過伺服器回送給客戶端。最常用來寫CGI程式的語言有：
· Unix 系統：Shell Script（sh、csh或ksh）、Perl、C/C++。
· Microsoft Windows 系統：DOS 批次檔（.bat）、Visual Basic、C/C++、可在Windows環境下執行的Perl版本，如BigPerl。

· Apple麥金塔系統：AppleScript、HyperCard、可在Mac執行的Perl版本，如MacPerl。
CGI雖然在Unix相容的作業平台上使用得相當普遍，但是在微軟的Windows作業環境中卻水土不服，效能低落。微軟為了解決這個問題乃提出ASP方案，應用物件導向的程式設計理念，利用內建的Request、Response、Server、Session、Application ... 等物件來運作，不再像CGI一樣依賴環境變數，使得程式的撰寫容易得多，執行的效率也高得多。再者，透過ADO物件，ASP程式可以很容易的存取各類型資料庫；外掛各式各樣的ActiveX，ASP程式和瀏覽器的功能便可不斷擴充。基於以上種種原因，ASP推出不久便得到廣泛的注意和使用。然而，ASP的缺點是它是微軟為自己的作業系統（Windows）和伺服器（IIS或PWS）所量身定作的，所以如果您的伺服器是在Unix相容的作業系統中執行，那麼ASP就英雄無用武之地了。

ASP程式也是由描述語言（Script）所構成的，JavaScript和VBScript是目前ASP程式常用的兩種描述語言。但是由於ASP程式是在伺服器端執行，所以它是屬於Server-end Script而非Client-end Script。

15-2 ASP基本語法


ASP原始碼（附檔名為 .asp）和HTML、JavaScript、VBScript（附檔名為 .htm或 .html）一樣都是文字檔，只不過HTML、JavaScript、VBScript文檔是將原始碼送至瀏覽器解譯、顯示，使用者直接便可使用瀏覽器檢視原始碼；ASP程式的原始碼則在Server端解譯，執行結果才送至瀏覽器顯示，因此使用者看到的不是ASP原始碼而是其執行結果。

撰寫ASP程式的基本步驟為：


1. 使用任何文書編輯器編輯原始碼；


2. 儲存成一般文字檔，附檔名使用 .asp；


3. 將該程式送至伺服器的特定目錄下（例如：c:\Inetpub\wwwroot）。

ASP標準語法如下：


<Script Language="VBScript" RUNAT="Server">



ASP程式碼


</Script>

也可以採用簡潔語法，將HTML原始碼和ASP程式碼交雜在一起：

HTML原始碼


<%



ASP程式碼


%>

HTML原始碼


所以ASP實際上是一個可以在Server端執行的Script！從以上語法可知所有的ASP程式碼都被一組特定標籤 <% ... %> 所刮住，內嵌在HTML原始碼中，底下這個範例說明了此一概念。

· 範例 15-1A（ch15-01A.asp）
本範例示範ASP程式碼如何嵌入HTML原始碼中。
· 原始碼：
<html>

<head>

<meta http-equiv="Content-Type" content="text/html; charset=big5">

<title>範例15-1A</title>

</head>

<body>

<h1>簡易ASP範例</h1>

<p>這一行是一般的HTML文字</p>
<%



response.write "<p>這一行是由ASP所輸出的文字</p>"

%>
<p>這一行，也是一般的HTML文字</p>

</body>

</html>
· 執行結果：

· 說明：

1. ASP程式碼由 <% ... %> 刮住，內嵌在HTML原始碼中。

2. ASP程式必須使用瀏覽器透過http方式連上伺服器才能執行，無法單獨使用瀏覽器執行。

3. ASP程式使用response.write將資料輸出至瀏覽器。
4. ASP程式採用VBScript為預設語法。


當然，如果您比較熟悉JavaScript，您也可以使用JavaScript來撰寫ASP程式。這時，就要採用以下的語法了：


<Script Language="JavaScript" RUNAT="Server">



ASP程式碼


</Script>
· 範例 15-1B（ch15-01B.asp）
本範例示範如何使用JavaScript來撰寫ASP程式。
· 原始碼：
<html>

<title>範例15-1B</title>

<body>

<script Language="JavaScript" RUNAT="Server">


response.write("<h1>使用JavaScript語法寫ASP</h1>");


response.write("<p>ASP程式也是一種Script程式</p>");


response.write("<p>不過它是一種Server-end Script</p>");


response.write("<p>VBScript是ASP預設的語法</p>");


response.write("<p>當然也可以使用JavaScript</p>");

</script>

</body>

</html>

· 執行結果：

15-3 ASP常用物件


ASP採用了物件導向的概念，內建了許多物件供程式設計師使用，以下介紹幾個比較常用的。

15-3-1 Response：輸出物件

Response是ASP的一個非常重要的物件，提供了許多屬性（Property）和方法（Method），此處介紹兩個比較常用的方法Write和Redirect：
Write方法

· 用途：

將資料輸出到網頁上。

· 語法：


Response.Write 要輸出的資料
· 精簡語法：
如果要輸出的資料非常簡短，可以將Response.Write精簡成如下語法內嵌在HTML原始碼中。

<% =要輸出的資料 %>

· 範例15-2（ch15-02.asp）
<html>

<head>

<title>範例15-2</title>

</head>

<body>

<h1>擲骰子遊戲</h1>

<%


randomize

%>

<p>您的點數是：<% = int(rnd*6+1) %>
</p>

<p>

請按「重新整理」便可重擲</p>

</body>

</html>

· 執行結果：

Redirect方法

· 用途：

將網頁轉向某個特定的網址。

· 語法：


Response.Redirect "網址URL"

· 範例 15-3（ch15-03.asp）
本範例示範如何依據不同條件將網頁轉向到不同的目的地。
· 程式碼：
<%

randomize

x = rnd * 10

if x < 5 then


response.redirect "ch15-03a.htm"

else


response.redirect "ch15-03b.htm"

end if

%>
· 執行結果：

15-3-2 Request：輸入物件

Request是ASP程式取得表單輸入資料的重要物件，此處介紹附屬於Response的兩個常用的屬性集（Collection）：Form和QueryString。

Form屬性集

· 用途：

取得表單以POST方法傳送的資料。

· 語法：


Request.Form("表單物件名稱")
範例 15-4（ch15-04.htm、ch15-04.asp）
本範例首先由ch15-04.htm送出一個表單給使用者，填妥資料後以POST方法回送給Server端的ch15-04.asp處理。

· 表單畫面原始碼：ch15-04.htm

<html>

<meta http-equiv="Content-Type" content="text/html; charset=big5">

<title>範例15-4</title>

<body>

<h1>範例：</h1>

<ul>

  <li>ch15-04.htm: 以 POST 方法傳送表單資料</li>

  <li>ch15-04.asp: 以 Request.Form 讀取表單資料</li>

</ul>

<hr> 

<p>表單畫面：</p>

<form method="post" action="ch15-04.asp">

<p>姓名: <input type="text" name="txtName">

<p>性別:

<input type="radio" name="rdoSex" value="男">帥哥

<input type="radio" name="rdoSex" value="女">美女

<p><input type="submit" name="B1" value="送出">
<input type="reset" name="B2" value="重填">
</form>

</body>

</html>
· 執行結果：

· 說明：

在HTML原始碼中（ch15-04.htm）以下述標籤指定表單資料必須以POST方式傳送給ch15-04.asp程式處理：
<form method="post" action="ch15-04.asp">

因此，使用者填妥資料後按下「送出」鈕，表單中的欄位資料便會被送回Server端由以下的ch15-04.asp程式來處理。
· ASP程式：ch15-04.asp

<html>

<head>

<title>範例15-04</title>

</head>

<body>

<h1>範例：</h1>

<ul>

  <li>ch15-04.htm: 以 POST 方法傳送表單資料</li>  

  <li>ch15-04.asp: 以 Request.Form 讀取表單資料</li>  

</ul>  

<hr> 

<p>  

您的資料為：</p>

<p> 

<% 


Response.Write "姓名: " & Request.Form("txtName") & "<br>" 


Response.Write "性別: " & Request.Form("rdoSex") 

%> 

<p> 

</body>

</html>
· 說明：
在此ASP程式中（ch15-04.asp），以Request.Form("txtName") 取得上述表單中的txtName資料，以Request.Form("rdoSex") 取得rdoSex資料，並以Response.Write把資料輸出為如下畫面。
· 執行結果：

QueryString屬性集
· 用途：

取得 (1) 表單以GET方法傳送的資料，或 (2) 伴隨在網址後面的「查詢字串」（Query String）。
· 語法：


Request. QueryString ("表單物件名稱")
或

Request.QueryString("引數名稱")
· 範例 15-5（ch15-05.htm、ch15-05.asp）
本範例首先由ch15-05.htm送出一個表單給使用者，填妥資料後以GET方法回送給Server端的ch15-05.asp處理。

· 表單畫面原始碼：ch15-04.htm

<html>

<meta http-equiv="Content-Type" content="text/html; charset=big5">

<title>範例15-5</title>

<body>

<h1>範例：</h1>

<ul>

  <li>ch15-05.htm: 以 GET 方法傳送表單資料</li>

  <li>ch15-05.asp: 以 Request.QueryString 讀取表單資料</li>

</ul>

<hr> 

<p>表單畫面：</p>

<form method="get" action="ch15-05.asp">

<p>姓名: <input type="text" name="txtName">

<p>性別:

<input type="radio" name="rdoSex" value="男">帥哥

<input type="radio" name="rdoSex" value="女">美女

<p><input type="submit" name="B1" value="送出">
<input type="reset" name="B2" value="重填">
</form>

</body>

</html>
· 執行結果：

· 說明：

在HTML原始碼中（ch15-05.htm）以下述標籤指定表單資料必須以GET方式傳送給ch15-05.asp程式處理：
<form method="get" action="ch15-05.asp">

因此，使用者填妥資料後按下「送出」鈕，表單中的欄位資料便會被送回Server端由以下的ch15-05.asp程式來處理。
ASP程式：ch15-05.asp

<html>

<head>

<title>範例15-05</title>

</head>

<body>

<h1>範例：</h1>

<ul>

  <li>ch15-05.htm: 以 GET 方法傳送表單資料</li>  

  <li>ch15-05.asp: 以 Request.QueryString 讀取表單資料</li>  

</ul>  

<hr> 

<p>  

您的資料為：</p>

<p> 

<% 


Response.Write "姓名: " & Request.QueryString("txtName") & "<br>" 


Response.Write "性別: " & Request. QueryString ("rdoSex") 

%> 

<p> 

</body>

</html>
· 說明：
在此ASP程式中（ch15-05.asp），以Request.QueryString("txtName") 取得上述表單中的txtName資料，以Request.QueryString("rdoSex") 取得rdoSex資料，並以Response.Write把資料輸出為如下畫面。
表單的GET方法和POST方法最大不同在於GET會將表單內各欄位的資料串成「查詢字串」（Query String）格式，以問號 ? 串接在網址後面送出：
網址?查詢字串
其中的「查詢字串」則是由各欄位的名稱作為引數名稱，欄位的內容作為引數值，每組引數之間以 & 作為區隔，如下：
引數名稱1=引數值1&引數名稱2=引數值2&...&引數名稱n=引數值n
· 執行結果：

註：畫面中看似亂碼的引數值%AAL%ABH%A6%A8，是因為中文編碼的結果。

由於Query String採用開放原則傳遞引數，所以資料輸入不一定要透過表單，只要遵循前述的格式，組成一個完整的查詢字串接在網址後面，即可將引數夾帶給ASP程式處理，如下圖所示。

另外，也可以利用超連結方式來夾帶查詢字串，如下所示：
<a href="http://網址?查詢字串> 超連結 </a>

15-3-3 Server：伺服器物件

Server是ASP的一個非常重要的物件，提供了許多屬性（Property）和方法（Method），此處僅介紹比較常用的方法CreateObject：
MapPath方法

· 用途：

傳回虛擬路徑所對應的實際路徑。

· 語法：

Server.MapPath("虛擬路徑")

CreateObject方法

· 用途：

在Server端的ASP程式中建立一個新的ActiveX物件。
· 語法：

Server.CreateObject("ActiveX物件名稱")

通常Server可以省略而簡寫成：

CreateObject("ActiveX物件名稱")


以下僅以「檔案系統物件」和「資料庫物件」來說明ASP程式搭配ActiveX物件的用法。
15-4檔案系統物件


ASP程式可利用檔案系統物件（File System Object），對伺服器端的檔案進行存取動作。

15-4-1建立檔案系統物件

· 物件名稱：
Scripting.FileSystemObject
· 建立語法：
Set fs = CreateObject("Scripting.FileSystemObject")
· 說明：

1. 這段指令會傳回一個FileSystemObject物件。
2. fs為自訂的變數名稱，用來接收傳回的FileSystemObject物件。

15-4-2文字檔的處理


我們可以利用檔案系統物件的OpenTextFile方法來處理文字檔。
開啟/建立文字檔案


由FileSystemObject的OpenText方法所建立的物件稱為FileTextStream物件，提供了ASP程式處理文字檔的重要管道。
· 語法：

Set ts = fsobj.OpenTextFile(filename, iomode, creat, format)
· 說明：

fsobj是任意的一個檔案系統物件。

OpenTextFile會傳回一個TextStream物件。
ts是自訂的變數名稱，此處用來接收傳回的TextStream物件。

filename為檔案名稱，必須是一個字串。

iomode為檔案的存取模式：

	常數值
	存取模式
	說明

	1
	讀取模式
	能從檔案中讀取資料

	2
	寫入模式
	能將資料寫入檔案並覆寫原資料

	8
	附加模式
	能將資料寫入檔案附加在原資料之後


creat為True時表示建立新檔。
format為檔案存取格式：

	常數值
	存取格式
	說明

	0
	ASCII
	以ASCII格式存取該檔

	-1
	Unicode
	以Unicode格式存取該檔

	-2
	預設格式
	以預設格式存取該檔


TextStream物件提供了許多存取文字檔的方法，以下僅說明常用的Close、Read、ReadLine、Write、WriteLine：

關閉文字檔

· 語法：

 tsobj.Close

讀取文字檔案
· 語法：
· 讀取特定長度的字元

tsobj.Read(讀取的字元數)

· 讀取一行文字

tsobj.ReadLine
寫入文字檔案

· 語法：

· 寫入資料不附加換行符號

tsobj.Write(要寫入的資料)

· 寫入資料附加換行符號

tsobj.WriteLine(要寫入的資料)

· 說明：

以上的tsobj皆代表TextStream物件。
· 範例 15-6（ch15-06.asp）

本範例以FileSystemObject物件和TextStream物件設計訪客計數器。

· 原始碼：
<html>

<head>

<title>範例15-6</title>

</head>

<body>

<h1>範例：訪客計數器</h1>

<%
'建立FileSystemObject

Set fs = CreateObject("Scripting.FileSystemObject")

'設定一些常數
ForReading = 1

ForWriting = 2

Path = Server.MapPath(".\")

FileName = Path & "\counter.txt"

CreatIt = True

'檔案若不存在則建立它並寫入1

if NOT fs.FileExists(FileName) then


Set ts = fs.OpenTextFile(FileName, ForWriting, CreatIt)


ts.WriteLine(1)


ts.Close

end if

'讀取資料

Set ts = fs.OpenTextFile(FileName, ForReading)

n = ts.ReadLine

ts.Close

'輸出
Response.Write "歡迎！您是本站第 " & n & " 位訪客"

'寫入資料

Set ts = fs.OpenTextFile(FileName, ForWriting)

ts.WriteLine(n+1)

ts.Close

Set fs = nothing

%>

</body>

</html>
· 說明：

1. CreateObject("Scripting.FileSystemObject")建立FileSystemObject。

2. Server.MapPath(".\")取得本程式所在的實際路徑。
3. FileExists 可用來判斷檔案存不存在。

4. 用OpenTextFile開啟檔案、建立檔案及設定檔案的存取方式。

5. 用ReadLin讀取資料；用WriteLine寫入資料。
6. 用Close關閉檔案。
7. Set fs = nothing 釋回FileSystemObject。
· 執行結果：

15-5資料庫物件

ASP程式可利用微軟所發展的ADO物件（ActiveX Data Objects）來存取資料庫。ADO物件是由許多有用的資料庫物件所構成的，其架構圖如下。本節僅介紹常用的Connection物件和Recordset物件。

15-5-1 Connection物件
物件用途說明
Connection物件是所有ADO架構的最上層，主要負責與資料庫建立連結，並提供從屬物件來操作資料庫。因此要存取資料庫一定要新建立Connection物件。
· 物件名稱：

ADODB.Connection
· 建立Connection物件的語法：

 Set conn = CreateObject("ADODB.Connection")

· 說明：

1. 這段指令會傳回一個Connection物件。
2. conn為自訂的變數名稱，用來接收傳回的Connection物件。
開啟資料庫連結
· Open方法：

有了Connection物件，便可利用它所提供的Open方法與特定資料庫建立實際的連結。

· 語法：
 conn.Open ConnectionString, UserID, Password

· 說明：

· ConnectionString

1. 若使用DSN，則ConnectionString可設定為ODBC中所定義的DSN（Data Source Name）名稱。

2. 若不使用DSN，以Access和SQL Server資料庫為例，其ConnectionString設定分別為：

Access資料庫：
driver={Microsoft Access Driver (*.mdb)};dbq=c:\mydb\test.mdb
SQL Server資料庫：

driver={SQL Server}; server=emls;uid=sy;pwd=hy;database=mydb
· UserID：使用者名稱
· Password：密碼
15-5-2 Recordset物件

物件用途說明

Recordset物件代表的是資料庫中的資料集合，它提供了許多操作資料庫的方法。
· 物件名稱：

ADODB.Recordset
· 建立Recordset物件的語法：

 Set rs = CreateObject("ADODB.Recordset ")

· 說明：

1. 這段指令會傳回一個Recordset物件。
2. rs為自訂的變數名稱，用來接收傳回的Recordset物件。
Recordset物件的重要方法

· 開啟/產生記錄集的方法：

Open：將Connection所連結的資料庫藉由特定SQL查詢句開啟，而產生一個記錄集物件。
 rs.Open SQL, conn, cursortype, locktype
其中：

SQL為一段SQL敘述；

conn為一個動作中的Connection物件；

currsortype為記錄集存取型態；

locktype為記錄集共享方式。
· 關閉記錄集的方法：

Close：關閉現有的記錄集物件。
· 移動記錄指標的方法：
MoveFirst：移到的第一筆記錄。
MoveNext：移到的下一筆記錄。
MovePrevious：移到的上一筆記錄。
MoveLast：移到最後一筆記錄。

· 搜尋記錄的方法：
Find：找尋特定記錄。

· 判別記錄指標位置的方法：
EOF：判斷是否到了檔案結束處。
BOF：判斷是否到了檔案啟始處。

· 增刪記錄的方法：
AddNew：新增資料。

Delete：刪除資料。
Update：修改資料。
· 範例 15-7（ch15-07.asp）
本範例說明如何利用ADO物件和ODBC資料來源連結，以存取Access資料庫 ch15-07.mdb。

· 資料庫內容：

· ODBC資料來源名稱（DSN）：book

由於以下的ASP程式碼要利用ADO物件，並透過ODBC來連結上述資料庫，因此我們必須先設定ODBC資料來源，步驟如下：

· 步驟一：開啟控制台，選擇ODBC資料來源。


[image: image15.wmf] 


· 步驟二：選擇「系統資料來源名稱」，按「新增」。

· 步驟三：選擇適合您的資料庫使用的驅動程式，此處為Microsoft Access Driver (*.mdb)，按「完成」。

· 步驟四：輸入資料來源名稱，本例為「book」，按「選取」。

[image: image18.wmf] 


· 步驟五：切換工作路徑到資料庫所在的資料夾，選取該檔後按「確定」回到「ODBC Microsoft Access設定」畫面，再按「確定」。

· 步驟六：回到「ODBC資料來源管理員」畫面後，可以看到多了一個項目book。

經過以上設定之後，便可以使用「book」這個資料來源名稱，來代表ch15-07.mdb這個需要Microsoft Access 驅動程式才能存取的資料庫。
以下就來看看ASP程式碼：
· 原始碼：
<html>

<head>

<title>範例15-7</title>

</head>

<body>

<%

Set conn = CreateObject("ADODB.Connection")

conn.Open "book"

'利用ODBC資料來源（DSN）連結資料庫

sql = "select * from [書目資料表]"

Set rs = CreateObject("ADODB.Recordset")

rs.Open sql, conn

%>

<table border>

<caption>書目資料庫</caption>

<thead>

<tr>

<th>書名</th>

<th>作者</th>

<th>出版社</TH>

</tr>

</thead>

<tbody>

<%

rs.MoveFirst

do while Not rs.eof

%>

<tr>

<td><% =rs.Fields("書名") %></td>

<td><% =rs.Fields("作者") %></td>

<td><% =rs.Fields("出版社") %></td>

</tr>

<%

rs.MoveNext

loop

%>

</tbody>

</table>

</body>

</html>
· 執行結果：

· 範例 15-8（ch15-08.asp）
本範例說明如何利用ADO物件，不需要ODBC的DSN設定，而直接利用Access驅動程式存取資料庫 ch15-08.mdb。

· 資料庫內容：

ch15-08.mdb資料庫內容基本上和ch15-07.mdb完全一樣。

· 原始碼：
<html>

<head>

<title>範例15-8</title>

</head>

<body>

<%

path = Server.MapPath(".\")

db = path & "\ch15-08.mdb"
con_str = "DBQ=" & db & ";Driver={Microsoft Access Driver (*.mdb)};"

'直接指定驅動程式和檔案路徑來連結資料庫
Set conn = CreateObject("ADODB.Connection")

conn.Open con_str 

sql = "select * from [書目資料表]"

Set rs = CreateObject("ADODB.Recordset")

rs.Open sql, conn

%>

本範例使用以下的ConnectionString直接開啟Access檔案：<br>

<% =con_str %> <hr>

<table border>

<caption>書目資料庫</caption>

<thead>

<tr>

<th>書名</th>

<th>作者</th>

<th>出版社</TH>

</tr>

</thead>

<tbody>

<%

rs.MoveFirst

do while Not rs.eof

%>

<tr>

<td><%=rs.Fields("書名")%></td>

<td><%=rs.Fields("作者")%></td>

<td><%=rs.Fields("出版社")%></td>

</tr>

<%

rs.MoveNext

loop

%>

</tbody>

</table>

</body>

</html>
· 執行結果：

15-6資料庫查詢範例

本節利用範例說明如何搭配HTML表單和ASP程式，完成一個簡易的網站資料庫的查詢功能。

範例 15-9（ch15-09.htm、ch15-09.asp）

在這個範例中，使用者可以在HTML的表單畫面點選要查詢的欄位，並輸入欄位資料，即可進行查詢。

· HTML表單：

<head>

<title>範例15-9</title>

</head>

<body>

<h1>範例：查詢表單</h1>

<p>請選擇查詢欄位，然後輸入欲查詢的字串：</p>

<form method="POST" action="ch15-09.asp">

  <p><select size="1" name="lstFieldName">

    <option>書名</option>

    <option>作者</option>

    <option>出版社</option>

  </select><input type="text" name="txtFieldValue" size="20"></p>

  <p><input type="submit" value="送出" name="B1">

  <input type="reset" value="清除" name="B2"></p>

</form>

<p>　</p>

</body>

</html>

· ASP程式碼：
<html>

<head>

<title>範例15-9</title>

</head>

<body>

<%

path = Server.MapPath(".\")

db = path & "\book.mdb"

con_str = "DBQ=" & db & ";Driver={Microsoft Access Driver (*.mdb)};"

Set conn = CreateObject("ADODB.Connection")

conn.Open con_str

parm = "[" & request.form("lstFieldName") & "]"

data = "'%" & request.form("txtFieldValue") & "%'"

sql = "select * from [書目資料表] where " & parm & " like " & data

Set rs = CreateObject("ADODB.Recordset")

rs.Open sql, conn

%>

<table border>

<caption>書目資料庫</caption>

<thead>

<tr>

<th>書名</th>

<th>作者</th>

<th>出版社</TH>

</tr>

</thead>

<tbody>

<%

if rs.eof then

%>


<tr>


<td colspan="3"><font color=blue>


抱歉！找不到您要的資料!</font>


</td>


</tr>

<%

else


rs.MoveFirst


do while Not rs.eof

%>

<tr>

<td><%=rs.Fields("書名")%></td>

<td><%=rs.Fields("作者")%></td>

<td><%=rs.Fields("出版社")%></td>

</tr>

<%


rs.MoveNext


loop


end if

%>

</tbody>

</table>

</body>

</html>

· 說明：

1. 表單標籤中以action="ch15-09.asp" 指定處理表單資料的程式。

2. ASP程式中以parm = request.form("lstFieldName") 取得使用者點選的欄位名稱，以data = request.form("txtFieldValue") 取得使用者輸入的欄位資料。

3. 將parm和data串在SQL查詢敘述之後，作為檢索條件："select * from [書目資料表] where " & parm & " like " & data
· 執行結果：

· 範例 15-10（ch15-10.htm、ch15-10.asp）

在這個範例中，使用者只要輸入要查詢的資料字串，即可進行跨欄位的綜合查詢。
· HTML表單：ch15-10.htm
<html>

<head>

<title>範例15-10</title>

</head>

<body>

<h1>範例：跨欄位綜合查詢</h1>

<p>請輸入欲查詢的字串：</p>

<form method="POST" action="ch15-10.asp">

  <p><input type="text" name="txtFieldValue" size="20"></p>

  <p><input type="submit" value="送出" name="B1">

  <input type="reset" value="清除" name="B2"></p>

</form>

<p>　</p>

</body>

</html>
· ASP程式碼：ch15-10.asp

<html>

<head>

<title>範例15-10</title>

</head>

<body>

<%

path = Server.MapPath(".\")

db = path & "\book.mdb"

con_str = "DBQ=" & db & ";Driver={Microsoft Access Driver (*.mdb)};"

Set conn = CreateObject("ADODB.Connection")

conn.Open con_str

parm1 = "[書名]"

parm2 = "[作者]"

parm3 = "[出版社]"

data = "'%" & request.form("txtFieldValue") & "%'"

sql = "select * from [書目資料表] where " _



& parm1 & " like " & data " or " _



& parm2 & " like " & data & " or " _



& parm3 & " like " & data

Set rs = CreateObject("ADODB.Recordset")

rs.Open sql, conn

%>

<table border>

<caption>書目資料庫</caption>

<thead>

<tr>

<th>書名</th>

<th>作者</th>

<th>出版社</TH>

</tr>

</thead>

<tbody>

<%

if rs.eof then

%>


<tr>


<td colspan="3"><font color=blue>


抱歉！找不到您要的資料!</font>


</td>


</tr>

<%

else


rs.MoveFirst


do while Not rs.eof

%>

<tr>

<td><%=rs.Fields("書名")%></td>

<td><%=rs.Fields("作者")%></td>

<td><%=rs.Fields("出版社")%></td>

</tr>

<%


rs.MoveNext


loop


end if

%>

</tbody>

</table>

</body>

</html>

· 說明：

1. 表單標籤中以action="ch15-10.asp" 指定處理表單資料的程式。

2. ASP程式中以data = request.form("txtFieldValue") 取得使用者輸入的欄位資料。

3. 將data分別與三個欄位做Like比較，最後結果再OR起來作為SQL檢索條件。
· 執行結果：

習題

1. 何謂ASP？與CGI的異同為何？
2. ASP物件中作為輸出用途的是          ，輸入用途的是         。

3. 試設計一ASP程式，輸出如下的九九乘法表。

4. 試利用ASP程式建立一個簡易的訪客留言版。
5. 試以Access建立一個「同學錄」資料庫，並設計一個ASP程式供他人透過網路查詢資料庫內容。

6. 某地區圖書館欲將館藏書目資料電腦化，請以Access建立一個「圖書目錄」資料庫，並設計一個ASP程式供讀者透過網路查詢資料庫內容。

