[image: image1.png]A 1~100% A

142 +..+100 =5050





[image: image21.wmf] 

[image: image22.wmf] 


[image: image23.wmf] 


[image: image24.jpg]


[image: image25.wmf] 




[image: image26.wmf] 



10

迴圈與分支
10-1前言

迴圈控制可以讓我們在程式中重複執行某些敘述。在ASP程式中可以使用VBScript提供的三種迴圈控制方式：
1. For ... Next
2. For Each … Next
3. Do ... Loop
4. While ... Wend
有時，我們在程式中必須強迫改變程式流程，比如說，強迫跳躍至某個敘述，或強迫脫離迴圈，這就有賴分支或跳躍敘述了。VBScript提供了幾個可以影響程式執行流程的分支敘述：

1. Exit For

2. Exit Do

3. Goto
4. End
本章重點在於說明迴圈與分支敘述的用法。
10-2 For迴圈

· 用途：

For迴圈通常用於將某一敘述區段重複執行，且重複次數是固定的場合。For 迴圈的重複次數，可以經由設定一個計數變數及其上下限來決定。
· 語法：

For 計數變數 = 初值 To 終值 [ Step 增量]



[敘述區段]


Next
· 說明：

當程式中碰到For敘述時，其執行步驟如下：
1. 將計數變數（counter）設定為初值（start value）。
2. 若增量為正，則判斷計數變數值是否小於終值（end value）；若增量為負，則判斷計數變數值是否大於終值。

3. 若上述判斷為真（True），則執行敘述區段（statement block）中的敘述，否則跳出迴圈。
4. 碰到Next敘述，把計數變數加上增量。
5. 重複步驟 2 到 4。
· 範例 10-1（ch10-01.asp）


試設計一程式計算1到100的整數和(1 + 2 + 3 + … + 100)。
· ASP程式碼：
<html>

<title>範例10-1</title>

<body>

<h1>計算1~100整數和</h1>

1 + 2 + ... + 100 = 

<%

    sum = 0

    For i = 1 To 100

        sum = sum + i

    Next

    Response.Write sum

%>

</body>

</html>
· 執行結果：

· 範例 10-2（ch10-02.asp）


試設計一程式計算1到100的奇數和 (1 + 3 + 5 + … + 99)。
· ASP程式碼：
<html>

<title>範例10-1</title>

<body>

<h1>計算1~100奇數和</h1>

1 + 3 + 5 + ... + 99 = 

<%

    sum = 0

    For i = 1 To 100 step 2

        sum = sum + i

    Next

    Response.Write sum

%>

</body>

</html>
· 執行結果：
[image: image2.png]ATEL1~1003 B

1+3+5+..+99=2500





· 範例 10-3（ch10-03.htm、ch10-03.asp）

試設計一表單及ASP程式，可以在表單中輸入兩個整數N、M，然後由ASP程式計算N到M的整數和。
· 表單輸入畫面原始碼：ch10-03.htm

<html>

<head>

<meta http-equiv="Content-Type" content="text/html; charset=big5">

<title>範例10-3</title>

</head>

<body>

<h1>範例：</h1>

<ul>

  <li>ch10-03.htm: 輸入N, M</li>

  <li>ch10-03.asp: 讀取N, M資料，計算N...M整數和</li>

</ul>

<hr> 

<form method="post" action="ch10-03.asp">

<p>請輸入N：<input type="text" name="txtN">

<p>請輸入M：<input type="text" name="txtM">

<p><input type="submit" name="B1" value="送出">

<input type="reset" name="B2" value="重填">

</form>

</body>

</html>
· ASP程式碼：ch10-03.asp
<html>

<title>範例10-3</title>
<body>

<h1>計算N~M整數和</h1>
<%

    n = CInt(Request.Form("txtN"))

    m = CInt(Request.Form("txtM"))

    sum = 0

    For i = n To m

        sum = sum + i

    Next

    Response.Write "N=" & n & "<br>"

    Response.Write "M=" & m & "<br>"

    Response.Write n & "+" & n+1 & "+ ... + " & m & "="

    Response.Write sum

%>

</body>

</html>
· 執行結果：
· 表單輸入畫面：

· ASP程式執行結果
[image: image4.png]FHEN~MEEH

N=20
M=166
204214 ... +166=13671




10-3 For Each迴圈

· 用途：

針對一個陣列或集合物件中的所有元素，重複執行一段敘述。
· 語法：


For Each 元素 In 群組


[敘述區段]


Next [元素]

· 說明：

1. 元素是一個變數，用來對應群組中的各個成員。

2. 群組是由許多個別成員所組成的陣列或物件集合。

3. For Each的運作方式：

(1) 首先，將群組中的第1個成員指定給元素變數，然後執行敘述區段中的敘述。

(2) 執行到Next時，回到For Each處。

(3) 將群組中的下一個成員指定給元素變數，然後再執行敘述區段中的敘述。

(4) 反覆執行步驟(2), (3)，直到群組中的成員依序指定完畢為止。

· 範例 10-4（ch10-04.htm、ch10-04.asp）

試設計一簡易問卷程式，在表單畫面填好問卷後，以ASP程式輸出問卷結果。
· 表單輸入畫面原始碼：

<html>

<head>

<title>範例10-4</title>

</head>

<body>

<h1>範例：簡易問卷</h1>

<ul>

  <li>ch10-04.htm: 問卷表單</li>  

  <li>ch10-04.asp: 輸出結果</li>  

</ul>  

<hr>

<p>表單輸入畫面：</p>

<form  method="post" action="ch10-04.asp">


姓名：<input type="text" name="txtName">


<p>性別：


<input type="radio" name="rdoSex" value="男">男


<input type="radio" name="rdoSex" value="女">女 


</p> 


<p>喜歡的科目（可複選）：


<input type="checkbox" name="chkSub" value="國文">國文


<input type="checkbox" name="chkSub" value="英文">英文


<input type="checkbox" name="chkSub" value="數學">數學


<input type="checkbox" name="chkSub" value="物理">物理


<input type="checkbox" name="chkSub" value="化學">化學


<input type="checkbox" name="chkSub" value="地理">地理


<input type="checkbox" name="chkSub" value="歷史">歷史


</p>


<p><input type="submit" value="送出">


<input type="reset" value="清除">


</p>

</form>

</body> 

</html>
· ASP程式碼
<html>

<head>

<title>範例10-4</title>

</head>

<body>

<h1>範例：簡易問卷</h1>

<ul>

  <li>ch10-04.htm: 問卷表單</li>  

  <li>ch10-04.asp: 輸出結果</li>  

</ul>  

<hr>

<p>ASP程式執行結果：</p>

<p>您輸入的資料為：</p>

<table border="1"> 

<% 


Response.Write "<tr><td>姓名</td>"


Response.Write "<td>" & Request.Form("txtName") & "</td></tr>"


Response.Write "<tr><td>性別</td>"


Response.Write "<td>" & Request.Form("rdoSex") & "</td></tr>"


Response.Write "<tr><td>喜歡的科目</td>"


For Each x In Request.Form("chkSub")



Response.Write "<td>" & x & "</td>"


Next

%>

</table>

</p> 

</body>

</html>

· 說明：

在表單中有一組複選式的核取方塊chkSub，其資料會以陣列方式傳送。因此，ASP程式以Request.Form("chkSub")所取得的資料，是由chkSub中被使用者所核取的項目所組成的陣列。所以若想知道陣列中的元素為何，可以使用For Each來取得。
· 執行結果
· 表單輸入畫面

· ASP程式執行結果

10-4 Do迴圈

雖然，For ... Next提供了重複執行迴圈的功能，但其執行次數是固定的，必須預先設定。有些時候，我們希望迴圈能不限次數的重複執行，直到某個條件被滿足為止，這就有賴 Do ... Loop 迴圈了。

Do ... Loop 的語法基本上有三種形式：

· 無條件式的Do迴圈

· 條件前測的Do迴圈

· 條件後測的Do迴圈

以下逐一說明。

10-4-1無條件式的Do迴圈

· 用途：


無條件式的Do … Loop會產生無窮迴圈。在ASP程式中使用無窮迴圈將造成伺服器一直處於執行該程式的狀態，除非使用者在其瀏覽器上按下「停止」，否則伺服器將鞠躬盡瘁，當機而後已，不可不慎。
· 語法：


Do


[敘述區段]


Loop

10-4-2條件前測的Do迴圈

· 用途：

當需要先判斷條件，再執行迴圈中的敘述區段時使用之。
· 語法：

Do [{While | Until} 條件]



[敘述區段]


Loop
· 說明：
1. 此種語法會先判斷條件符不符合，再決定要不要執行迴圈中的敘述區段。

2. 判斷條件的方式可以使用While或Until，兩者為互補關係。

3. 使用While判斷句時，如果條件為True，則反覆執行迴圈中的敘述區段，直到條件變為False為止。

4. 反之，使用Until判斷時，只有當條件為False時，迴圈中的敘述區段才會被反覆執行，直到條件變為True為止。

範例 10-5（ch10-05.asp）
試用Do While ... Loop 方式設計一程式，計算1到100的整數和（1 + 2 + 3 + … + 100）。
· ASP程式碼：

<html>

<head><title>範例10-5</title></head>

<body>

<h1>範例：</h1>

<h3>以 Do While... Loop 計算1~100整數和</h3>

1 + 2 + ... + 100 = 

<%

    sum = 0

    i = 1

    Do While i <= 100

        sum = sum + i

        i = i + 1

    Loop

    Response.Write sum

%>

</body>

</html>
· 執行結果：

· 範例 10-6（ch10-06.asp）
試用Do While ... Loop 方式設計一程式，計算1到100的奇數和（1 + 3 + 5 + … + 99）。
· ASP程式碼：

<html>

<title>範例10-5</title>

<body>

<h1>範例：</h1>

<h3>以 Do While... Loop 計算1~100奇數和</h3>

1 + 3 + 5 + ... + 99 = 

<%

    sum = 0

    i = 1

    Do While i <= 100

        sum = sum + i

        i = i + 2

    Loop

    Response.Write sum

%>

</body>

</html>
· 執行結果：

· 範例 10-7（ch10-07.htm、ch10-07.asp）

試以Do Until … Loop方式設計一程式，輸入兩個整數N、M，並計算N到M的整數和。
· HTML表單原始碼：
<html>

<head>

<meta http-equiv="Content-Type" content="text/html; charset=big5">

<title>範例10-7</title>

</head>

<body>

<h1>範例：</h1>

<ul>

  <li>ch10-07.htm: 輸入N, M</li>

  <li>ch10-07.asp: 以 Do ... Loop 計算 N...M 整數和</li>

</ul>

<hr> 

<form method="post" action="ch10-07.asp">

<p>請輸入N：<input type="text" name="txtN">

<p>請輸入M：<input type="text" name="txtM">

<p><input type="submit" name="B1" value="送出">

<input type="reset" name="B2" value="重填">

</form>

</body>

</html>
· ASP程式碼：
<html>

<head>

<title>範例10-7</title>

</head>

<body>

<h1>以 Do ... Loop 方式計算 N~M 整數和</h1>
<%

    n = CInt(Request.Form("txtN"))

    m = CInt(Request.Form("txtM"))

    sum = 0

    i = n

    Do Until i > M

        sum = sum + i

        i = i + 1

    Loop

    Response.Write "N=" & n & "<br>"

    Response.Write "M=" & m & "<br>"

    Response.Write n & "+" & n+1 & "+ ... + " & m & "="

    Response.Write sum

%>

</body>

</html>
· 執行結果：
· 表單輸入畫面

· ASP程式執行結果

10-4-3條件後測的Do迴圈

· 用途：

當需要先執行迴圈中的敘述區段，再判斷條件時使用之。
· 語法：

Do



[敘述區段]


Loop [{While | Until} 條件]
· 說明：
1. 此語法會無條件的先執行迴圈中的敘述區段一次，再判斷Loop後面的條件符不符合，以決定要不要重複執行迴圈。

2. 判斷條件的方式可以使用While或Until，兩者為互補關係。

3. 使用While判斷句時，如果條件為True，則反覆執行迴圈中的敘述區段，直到條件變為False為止。

4. 反之，使用Until判斷時，只有當條件為False時，迴圈中的敘述區段才會被反覆執行，直到條件變為True為止。

10-5 While迴圈
· 用途
Do 迴圈有各種形式：Do ... Loop、Do While ... Loop、Do Until ... Loop、Do ... Loop Wile ...、Do ... Loop Until ...，其中最常用的便是Do While ... Loop，此一語法可以用While ... Wend來取代，兩者完全等效，也就是只要指定的條件式為True，就會重複執行迴圈中的敘述區段。
· 語法

While 條件



[敘述區段]

Wend
· 說明
如果條件為True，則所有敘述區段內的陳述式都會執行，一直執行到Wend陳述式。然後再回到While陳述式，並再一次檢查條件，如果條件還是為True，則重複此步驟；如果條件不為True，則程式會從Wend陳述式之後的指令行繼續執行。 

10-6巢狀迴圈
不論是For迴圈、For Each迴圈、Do迴圈或While迴圈，多個迴圈可以各自獨立、互不相干，也可以組成多層的巢狀結構，形成所謂的「巢狀迴圈」。巢狀迴圈的最大特色是互相套疊，內迴圈必然被完全包在外迴圈之中，如圖所示即為以For迴圈為例說明巢狀迴圈的觀念。

10-6-1巢狀For迴圈

· 語法：

For 計數變數_1 = 初值_1 To 終值_1 Step 增量_1

[敘述區段_1]


For 計數變數_2 = 初值_2 To 終值_2 Step 增量_2


[敘述區段_2]



…..



For 計數變數_n = 初值_n To 終值_n Step 增量_n



[敘述區段_n]



Next 計數變數_n



…..



[敘述區段_2]


Next 計數變數_2


[敘述區段_1]

Next 計數變數_1
· 範例 10-8（ch10-08.asp）

試設計一ASP程式，在網頁上輸出如下圖的九九乘法表。

· 執行畫面：

[image: image12.png]Hil : DUSURForiBl A 2 SRR

[*1=1[1%2=2 [[1¥3=3 [1*4=4 [1*5=5 [1%6=6 [1¥7=7 [1*8=8 [1*9=9
*1=2[2%2=4 |p¥3=6 |24 [2*5-=10[p*6=12[2*7=142%8=16[2*9=18
B*1=3326 [3%3=0 |[3¥4=123%5=153%6=183*7=21 5824 [3*9=27
[4F1=4 4728 [473=12[4%4=16 4¥5-00 476244728 4783247936
[5+1=55+2=10/5%3=15 5*4=20 5%5-25 [5%6=30[5*7=35 [5*8=40 [5*9=45
(6126 6%2=12 6318 [6*4=24 [6%5=30/6*6=36 6" 7=42 [6*8=48 [6*9=54
(7127 [7%2=14 #3221 [7%428 [1+5=35 7642 7743 [1°856 1963
[B+1=8]8*2=168*3=246*4 =32 8*5=40 6648 8 1=56 "8 64 [79=12
[6%1=90%2=189*3=27[0%4=36 [9¥5=45 [9%6=54 [0¥763 9%8 729951





· ASP程式碼：

<html>

<head>

<meta http-equiv="Content-Type" content="text/html; charset=big5">

<title>範例10-8</title>

</head>
<body>

<h1>範例：以巢狀For迴圈產生九九乘法表</h1>

<%

  Dim i, j, k

  Response.Write "<table border=1>"

  For i=1 to 9

  Response.Write "<tr>"

    For j=1 to 9

  
k = i * j

  
Response.Write "<td>" & i & "*" & j & "=" & k & "</td>"

    Next

    Response.Write "</tr>"

  Next

  Response.Write "</table>"

%>

</body>

</html>
· 說明：

1. 外圈的i迴圈每執行1次，內圈的j迴圈就要執行9次。

2. 程式中使用HTML的表格標籤群，如<table>...</table>、<tr>...</tr>、<td>...</td>等，以便有效的組織所顯示的資訊。

10-6-2巢狀Do迴圈（條件前測）

· 語法：


Do [{While | Until} 條件_1]



[敘述區段_1]



Do [{While | Until} 條件_2]




[敘述區段_2]




…..




Do [{While | Until} 條件_n]





[敘述區段_n]




Loop




…..




[敘述區段_2]



Loop



[敘述區段_1]


Loop
· 範例 10-9（ch10-09.asp）

試以Do … Loop重新設計上一個ASP程式，在網頁上輸出九九乘法表，如下圖所示。

· 執行畫面：

· ASP程式碼：

<html>

<head>

<meta http-equiv="Content-Type" content="text/html; charset=big5">

<title>範例10-9</title>

</head>

<body>

<h1>範例：以巢狀Do迴圈產生九九乘法表</h1>

<%

  Dim i, j, k

  Response.Write "<table border=1>"

  i = 1

  Do While i<=9

    Response.Write "<tr>"

    j = 1

    Do While j<=9


k = i * j

  
Response.Write "<td>" & i & "*" & j & "=" & k & "</td>"

  
j = j + 1

    Loop

    Response.Write "</tr>"

    i = i + 1

  Loop

  Response.Write "</table>"

%>

</body>

</html>
10-6-3巢狀Do迴圈（條件後測）

· 語法：


Do



[敘述區段_1]



Do



[敘述區段_2]




…..




Do




[敘述區段_n]




Loop [{While | Until} 條件_n]



…..




[敘述區段_2]



Loop [{While | Until} 條件_2]


[敘述區段_1]


Loop [{While | Until} 條件_1]

10-6-4巢狀While迴圈
· 語法：


While 條件_1


[敘述區段_1]



While 條件_2



[敘述區段_2]




…..




While 條件_n




[敘述區段_n]




Wend




…..




[敘述區段_2]



Wend



[敘述區段_1]


Wend
10-6-5巢狀混合迴圈

在巢狀迴圈中可以有For迴圈、Do迴圈、While迴圈或其他形式的迴圈混合出現。

· 範例 10-10（ch10-01.asp）

· 程式功能：


試設計一程式，輸出如下圖的梯形塔。
· 執行畫面：

· ASP程式碼：

<html>

<head>

<meta http-equiv="Content-Type" content="text/html; charset=big5">

<title>範例10-10</title>

</head>

<body>

<h1>範例：</h1>

<h2>以混合巢狀迴圈產生梯形塔</h2>

<%

  Dim i, j

  i = 1

  Response.Write "<table border>"

  Do

    Response.Write "<tr>"

    For j = 1 To i

      Response.Write "<td>O</td>"

    Next

    Response.Write "</tr>"

    i = i + 1

  Loop While i <= 10

  Response.Write "</table>"

%>

</body>

</html>
10-7分支跳躍

流程控制（判斷與選擇）和迴圈控制，都可用來改變ASP程式執行的順序，其執行的規則是根據程式師設定的條件判斷式，來決定程式流程如何進行。但有時，我們在程式中必須強迫改變程式流程。比如說，強迫跳躍至某個敘述，或強迫脫離迴圈。VBScipt提供了幾個敘述，可用於ASP程式中，作為無條件的分支跳躍之用：

· Exit For

· Exit Do

· Exit Sub

· Exit Function

· Goto
以下就一一介紹其用法。
10-7-1 Exit For

· 用途：
強迫脫離For ... Next迴圈。

· 說明：

一般而言，For迴圈執行的次數取決於計數變數的初值、終值和增量。但如果在 For迴圈中加入Exit For敘述，將會強迫程式跳出該迴圈，而不論計數變數的值是多少。
10-7-2 Exit Do

· 用途：

強迫脫離Do ... Loop迴圈。

· 說明：

一般而言，Do迴圈執行與否取決於While或Until條件判斷式是否滿足。但如果在Do ... Loop迴圈中加入Exit Do敘述，將會強迫程式跳出該迴圈，而不論條件是否滿足。
10-7-3 Exit Sub

· 用途：

強迫結束Sub程序。

· 說明：

在ASP程式中，我們可以使用Sub程序（Sub procedure）寫一段副程式，以供主程式在需要時呼叫。Sub程序的程式碼皆置於Sub … End Sub之間，其語法為：


[Private | Public] Sub 程序名稱([引數串列])



[敘述區段_1]


[Exit Sub]



[敘述區段_2]

End Sub
一般而言，Sub程序必須執行到End Sub才會結束，但如果在敘述區段中加入Exit Sub敘述，則會強迫Sub程序結束。
10-7-4 Exit Function

· 用途：

強迫結束Function函式。

· 說明：

在ASP程式中，我們可以使用Function程序（Function procedure）寫一段函式，以供主程式在需要時呼叫。Function和Sub最大的不同在於Function必需有回傳值（Return Value），而Sub則無。Function程序的程式碼皆置於Function … End Function之間，其語法為：


[Private | Public] Function 函式名稱([引數串列])



[敘述區段_1]


[Exit Sub]



[敘述區段_2]

End Function

一般而言，Function程序必須執行到End Function才會結束，但如果在敘述區段中加入Exit Function敘述，則會強迫Function程序結束。

習題
1. 試設計一ASP程式，隨機產生10個數，輸出原始數列及最大數。

2. 試設計一ASP程式，隨機產生10個數，輸出原始數列及最小數。
3. 試設計ASP程式，隨機產生10個數，輸出原始數列及平均值。
4. 試設計一ASP程式，輸出下列結果：

5. 試設計一ASP程式，計算下列級數和。

(A) [image: image18.bmp]
(B) [image: image19.bmp]
(C) 先輸入N，再求[image: image20.bmp]
6. 假設銀行定期存款年利率為N%，某甲現有本金M元，若以複利方式存入該銀行，試設計一HTML表單供使用者輸入N和M，再以一ASP程式計算並輸出1~10年期定存的本利和。

7. 試設計一ASP程式，隨機產生一正整數N，找出N的所有因數並輸出。
8. 試設計一ASP程式，隨機產生一正整數N，求小於N的所有質數並輸出。

9. 因數分解：試設計一ASP程式，隨機產生一正整數N，並將其分解成質因數的乘積，例如N = 24 = 2*2*2*3。
10. 試設計一ASP程式，隨機產生兩正整數M和N，求其最大公因數及最小公倍數。

